Fermi National Accelerator Laboratory

FERMILAB-TM-1781

A Generalized TRL Algorithm for S-Parameter De-Embedding

P. Colestock and M. Foley

Fermi National Accelerator Laboratory
P.O. Box 500, Batavia, Illinois 60510

April 1993

Operated by Universities Research Association Inc. under Contract No. DE-AC02-76CHO3000 with the United States Department of Energy

A GENERALIZED TRL ALGORITHM FOR S-PARAMETER DE-EMBEDDING

P. Colestock
Fermi National Accelerator Laboratory

M. Foley
Fermi National Accelerator Laboratory

FNAL Technical Memo TM-1781

Abstract

At FNAL bench measurements of the longitudinal impedance of various beamline components have been performed using stretched wire methods. The basic approach is to use a network analyzer (NWA) to measure the transmission and reflection characteristics (s-parameters) of the beam line component. It is then possible to recover the effective longitudinal impedance from the s-parameters. Several NWA calibration procedures have been implemented in an effort to improve the accuracy of these measurements. These procedures are mathematical techniques for extracting the s-parameters of a test device from external NWA measurements which include the effect of measurement fixtures. The TRL algorithm has proven to be the most effective of these techniques. This method has the advantage of properly accounting for the nonideal calibration standards used in the NWA measurements.

1 Introduction

The objective of this work is to recover an equivalent impedance for a given device-under-test (DUT) using a bi-directional reflectometer, otherwise known as a network analyzer (NWA). The basic algorithm consists of applying an incident wave to the DUT, which is characterized as a general two-port network, and measuring the vector voltages scattered into the forward and reverse directions. The resulting data can be used to calculate s-parameters. However, the measurements are complicated by the fact that transitions occur between the NWA and the DUT. The diagram below is a schematic representation of the measurement setup. A and B are general, linear networks representing the errors occuring in the s-parameter measurements of the DUT. The influence of error networks A and B must be removed from the data in order to accurately evaluate the s-parameters of the DUT. Using standard circuit analysis, it is possible to recover the effective longitudinal, as well as transverse, impedance of the DUT from the de-embedded s-parameters.



The method described in this Technical Memo is based on a generalization of the Thru-Reflect-Line (TRL) algorithm [1,2]. The calibration standards required are two lengths of transmission line and two shorts with equal reflection coefficient. The lengths of the transmission lines and the value of the reflection coefficient for the shorts are not required to be known. However, the ratio of the lengths of the two transmission lines is required.

Assuming the transmission lines used for calibration are nonreflecting, the s-parameter matrices for line 1 and line 2 are defined by

$$[S_{L1}] = \begin{bmatrix} 0 & L_1^+ \\ L_1^+ & 0 \end{bmatrix} \tag{1}$$

and

$$[S_{L2}] = \begin{bmatrix} 0 & L_2^+ \\ L_2^+ & 0 \end{bmatrix} \tag{2}$$

The s-parameter matrix for both shorts is

$$[S_{SHORT}] = \begin{bmatrix} \gamma & 0 \\ 0 & \gamma \end{bmatrix} \tag{3}$$

2 Derivation of Equations Relating S-Parameters Measured at NWA Ports to S-Parameters of Test Device

The object of this section is to find the expressions which relate the s-parameters of the DUT, S_{ij}, to the NWA measurements, S_{ijm}, where it is assumed that the s-parameters of the networks A and B have been determined, including the complex phase factor L_1^+. In Section 4 the expressions associated with the TRL calibration method which yield these network s-parameters are derived.

The network flow graph for the generalized TRL calibration is shown in Figure 1. The reference planes for this calibration method are located at the middle of the shorter transmission line. Therefore, half the length of the shorter line is included on each side of the DUT.

In order to develop expressions relating the s-parameters measured at the NWA ports, S_{ijm}, to the s-parameters of the DUT, S_{ij}, one follows the procedure of [3,4]. From the network flow graph in Figure 1:

$$\begin{aligned}
b_0 &= S_{11A}a_0 + S_{12A}a_1 && (4)\\
b_1 &= S_{21A}a_0 + S_{22A}a_1 && (5)\\
a_1 &= L_1^+ S_{11}b_1 + L_1^+ S_{12}b_2 && (6)\\
a_2 &= L_1^+ S_{21}b_1 + L_1^+ S_{22}b_2 && (7)\\
b_2 &= S_{11B}a_2 + S_{12B}a_3 && (8)\\
b_3 &= S_{21B}a_2 + S_{22B}a_3 && (9)
\end{aligned}$$

By definition, the s-parameters measured at the NWA ports are:

$$S_{11m} = \frac{b_0}{a_0}\,|_{a_3=0}$$



Figure 1: Network Flow Graph for Generalized TRL Calibration

and

$$S_{21m} = \frac{b_3}{a_0} \mid_{a_3=0}$$

From (8) and (9), for S_{11m} and S_{21m} $(a_3 = 0)$.

$$b_2 = S_{11B} a_2 \tag{10}$$

$$b_3 = S_{21B} a_2 \tag{11}$$

Therefore,

$$a_2 = \frac{b_3}{S_{21B}} \tag{12}$$

From (10) and (12),

$$b_2 = \frac{S_{11B}}{S_{21B}} b_3 \tag{13}$$

From (6)and (13),

$$a_1 = L_1^+ S_{11} b_1 + L_1^+ S_{12} \frac{S_{11B}}{S_{21B}} b_3 \tag{14}$$

From (5) and (14),

$$a_1 = L_1^+ S_{11} \left(S_{21A} a_0 + S_{22A} a_1 \right) + L_1^+ S_{12} \frac{S_{11B}}{S_{21B}} b_3 \tag{15}$$

and

$$a_1(1 - L_1^+ S_{11} S_{22A}) = L_1^+ S_{11} S_{21A} a_0 + L_1^+ S_{12} \frac{S_{11B}}{S_{21B}} b_3 \tag{16}$$

From (4) and (16),

$$\begin{aligned}(1 - L_1^+ S_{11} S_{22A}) b_0 &= S_{11A}(1 - L_1^+ S_{11} S_{22A}) a_0 + \\ &\quad L_1^+ S_{11} S_{12A} S_{21A} a_0 + L_1^+ S_{12} \frac{S_{12A} S_{11B}}{S_{21B}} b_3 \end{aligned} \tag{17}$$

Dividing through both sides of (17) by a_0 yields,

$$\begin{aligned}(1 - L_1^+ S_{11} S_{22A}) S_{11m} &= S_{11A}(1 - L_1^+ S_{11} S_{22A}) + \\ &\quad L_1^+ S_{11} S_{12A} S_{21A} + L_1^+ S_{12} \frac{S_{12A} S_{11B}}{S_{21B}} S_{21m} \end{aligned} \tag{18}$$

and

$$\begin{aligned} S_{11m} &= S_{11A} + L_1^+ S_{11}(S_{12A} S_{21A} + S_{22A} S_{11m} - S_{11A} S_{22A}) + \\ &\quad L_1^+ S_{12} \frac{S_{12A} S_{11B}}{S_{21B}} S_{21m} \end{aligned} \tag{19}$$

Multiplying the last term in (19) by $\frac{L_1^+ S_{21A}}{L_1^+ S_{21A}}$ and simplifying, one obtains,

$$\begin{aligned} &(S_{11m} - S_{11A}) S_{21A} S_{21B} L_1^+ = \\ &\quad (S_{12A} S_{21A} L_1^+ + S_{22A} L_1^+ S_{11m} - S_{11A} S_{22A} L_1^+) S_{21A} S_{21B} L_1^+ S_{11} + \\ &\quad S_{12A} S_{21A} L_1^+ S_{11B} L_1^+ S_{21m} S_{12} \end{aligned} \tag{20}$$

In order to determine S_{22m} in terms of the s-parameters of the DUT, make the following substitutions in (20):

Replace	*By*
S_{11A}	S_{22B}
S_{12A}	S_{21B}
S_{21A}	S_{12B}
S_{22A}	S_{11B}
S_{11B}	S_{22A}
S_{21B}	S_{12A}
S_{11m}	S_{22m}
S_{21m}	S_{12m}
S_{11}	S_{22}
S_{12}	S_{21}

Equation (20) becomes,

$$\begin{aligned}
&(S_{22m} - S_{22B})S_{12A}S_{12B}L_1^+ = \\
&\quad (S_{12B}S_{21B}L_1^+ + S_{11B}L_1^+S_{22m} - S_{11B}S_{22B}L_1^+)S_{12A}S_{12B}L_1^+S_{22} + \\
&\quad S_{12B}S_{21B}L_1^+S_{22A}L_1^+S_{12m}S_{21}
\end{aligned} \tag{21}$$

From (4),

$$S_{22A}b_0 = S_{22A}S_{11A}a_0 + S_{22A}S_{12A}a_1 \tag{22}$$

From (5),

$$S_{12A}b_1 = S_{12A}S_{21A}a_0 + S_{12A}S_{22A}a_1 \tag{23}$$

From (22) and (23),

$$S_{22A}b_0 - S_{22A}S_{11A}a_0 = S_{12A}b_1 - S_{12A}S_{21A}a_0 \tag{24}$$

From (24),

$$S_{22A}b_0 + (S_{12A}S_{21A} - S_{11A}S_{22A})a_0 = S_{12A}b_1 \tag{25}$$

From (25),

$$b_1 = \frac{S_{22A}}{S_{12A}}b_0 + \left(S_{21A} - \frac{S_{11A}S_{22A}}{S_{12A}}\right)a_0 \tag{26}$$

From (26) and (7),

$$a_2 = L_1^+S_{21}\frac{S_{22A}}{S_{12A}}b_0 + L_1^+S_{21}\left(S_{21A} - \frac{S_{11A}S_{22A}}{S_{12A}}\right)a_0 + L_1^+S_{22}b_2 \tag{27}$$

From (27) and (10),

$$\begin{aligned}
(1 - L_1^+S_{11B}S_{22})a_2 \quad &= \quad L_1^+S_{21}\frac{S_{22A}}{S_{12A}}b_0 + \\
&\quad\quad L_1^+S_{21}\left(S_{21A} - \frac{S_{11A}S_{22A}}{S_{12A}}\right)a_0
\end{aligned} \tag{28}$$

From (28) and (12),

$$\begin{aligned}
\frac{(1 - L_1^+S_{11B}S_{22})}{S_{21B}}b_3 \quad &= \quad L_1^+S_{21}\frac{S_{22A}}{S_{12A}}b_0 + \\
&\quad\quad L_1^+S_{21}\left(S_{21A} - \frac{S_{11A}S_{22A}}{S_{12A}}\right)a_0
\end{aligned} \tag{29}$$

Dividing through both sides of (29) by a_0 yields,

$$\begin{aligned}\frac{(1 - L_1^+ S_{11B} S_{22})}{S_{21B}} S_{21m} &= L_1^+ S_{21} \frac{S_{22A}}{S_{12A}} S_{11m} + \\ &\quad L_1^+ S_{21} \left(S_{21A} - \frac{S_{11A} S_{22A}}{S_{12A}} \right) \end{aligned} \tag{30}$$

and

$$\begin{aligned} S_{21m} &= L_1^+ S_{21} S_{21A} S_{21B} \left(1 + \frac{S_{22A}}{S_{12A} S_{21A}} S_{11m} - \frac{S_{11A} S_{22A}}{S_{12A} S_{21A}} \right) + \\ &\quad L_1^+ S_{22} S_{11B} S_{21m} \end{aligned} \tag{31}$$

Multiplying both sides of (31) by $S_{12A} s_{21A} L_1^+$ yields,

$$\begin{aligned} & S_{21m} S_{12A} S_{21A} L_1^+ = \\ & \quad (S_{12A} S_{21A} L_1^+ + S_{22A} L_1^+ S_{11m} - S_{11A} S_{22A} L_1^+) S_{21A} S_{21B} L_1^+ S_{21} + \\ & \quad S_{12A} S_{21A} L_1^+ S_{11B} L_1^+ S_{21m} S_{22} \end{aligned} \tag{32}$$

In order to determine S_{12m} in terms of the s-parameters of the DUT, make the same substitutions as before in (32):

$$\begin{aligned} & S_{12m} S_{12B} S_{21B} L_1^+ = \\ & \quad (S_{12B} S_{21B} L_1^+ + S_{11B} L_1^+ S_{22m} - S_{11B} S_{22B} L_1^+) S_{12A} S_{12B} L_1^+ S_{12} + \\ & \quad S_{12B} S_{21B} L_1^+ S_{22A} L_1^+ S_{12m} S_{11} \end{aligned} \tag{33}$$

Equations (20), (21), (32) and (33) relate the s-parameters measured at the NWA ports, S_{ijm}, to the s-parameters of the DUT, S_{ij}.

3 Standard NWA Error Model

In order to find the s-parameters of the error networks A and B, it is useful to define a set of error terms which represent forward and reverse coupling factors at each network.

The network flow graph of error terms for the generalized TRL calibration is shown in Figure 2. The corresponding error terms are given by [2]:



Forward Error Model

Reverse Error Model

Figure 2: Network Error Model for Generalized TRL Calibration

$$
\begin{array}{ll}
E_{df} = S_{11A} & E_{dr} = S_{22B} \\
E_{sf} = S_{22A}L_1^+ & E_{sr} = S_{11B}L_1^+ \\
E_{rf} = S_{12A}S_{21A}L_1^+ & E_{rr} = S_{12B}S_{21B}L_1^+ \\
E_{lf} = S_{11B}L_1^+ & E_{lr} = S_{22A}L_1^+ \\
E_{tf} = S_{21A}S_{21B}L_1^+ & E_{tr} = S_{12A}S_{12B}L_1^+
\end{array}
$$

Using the error terms defined above, equations (20), (21), (32) and (33) become

$$
\begin{aligned}
(S_{11m} - E_{df})E_{tf} &= (E_{rf} + E_{sf}S_{11m} - E_{df}E_{sf})E_{tf}S_{11} + \\
&\quad E_{rf}E_{lf}S_{21m}S_{12}
\end{aligned} \tag{34}
$$

$$
\begin{aligned}
S_{21m}E_{rf} &= (E_{rf} + E_{sf}S_{11m} - E_{df}E_{sf})E_{tf}S_{21} + \\
&\quad E_{rf}E_{lf}S_{21m}S_{22}
\end{aligned} \tag{35}
$$

$$
\begin{aligned}
(S_{22m} - E_{dr})E_{tr} &= (E_{rr} + E_{sr}S_{22m} - E_{dr}E_{sr})E_{tr}S_{22} + \\
&\quad E_{rr}E_{lr}S_{12m}S_{21}
\end{aligned} \tag{36}
$$

$$
\begin{aligned}
S_{12m}E_{rr} &= (E_{rr} + E_{sr}S_{22m} - E_{dr}E_{sr})E_{tr}S_{12} + \\
&\quad E_{rr}E_{lr}S_{12m}S_{11}
\end{aligned} \tag{37}
$$

4 Calculation of Error Terms

The purpose of this section is to evaluate the error terms defined in the preceding section using a set of calibration standards. First a transmission line is connected between the networks A and B, and a set of s-parameters are measured at the NWA ports. Then a second transmission line with a known incremental length relative to line 1 is connected and the measurements are repeated. Third, a short with an unknown reflection coefficient is connected at each network in turn, and the reflection coefficients at the NWA are measured. The relevant expressions which yield the error terms defined above are derived in this section.

For a general two-port network of the form

define the *wave cascade matrix* [R] by

$$\begin{bmatrix} b_1 \\ a_1 \end{bmatrix} = [R] \begin{bmatrix} a_2 \\ b_2 \end{bmatrix} = \begin{bmatrix} R_{11} & R_{12} \\ R_{21} & R_{22} \end{bmatrix} \begin{bmatrix} a_2 \\ b_2 \end{bmatrix} \tag{38}$$

Note that, in terms of the s-parameters of the two-port network,

$$[R] = \frac{1}{S_{21}} \begin{bmatrix} -\Delta & S_{11} \\ -S_{22} & 1 \end{bmatrix} \tag{39}$$

where $\Delta = S_{11}S_{22} - S_{12}S_{21}$.

If the wave cascade matrices of the error boxes A and B are denoted by $[R_A]$ and $[R_B]$ respectively, and those for line 1 and line 2 are $[R_{L1}]$ and $[R_{L2}]$, then successively connecting line 1 and line 2 between error boxes A and B yields,

$$[R_{D1}] = [R_A][R_{L1}][R_B] \tag{40}$$

$$[R_{D2}] = [R_A][R_{L2}][R_B] \tag{41}$$

Note that from (1), (2) and (39),

$$[R_{L1}] = \begin{bmatrix} L_1^+ & 0 \\ 0 & 1/L_1^+ \end{bmatrix} = \begin{bmatrix} L_1^+ & 0 \\ 0 & L_1^- \end{bmatrix} \tag{42}$$

and

$$[R_{L2}] = \begin{bmatrix} L_2^+ & 0 \\ 0 & 1/L_2^+ \end{bmatrix} = \begin{bmatrix} L_2^+ & 0 \\ 0 & L_2^- \end{bmatrix} \tag{43}$$

Eliminating $[R_B]$ from (40) and (41), one obtains

$$\begin{aligned} [R_{D2}][R_{D1}]^{-1}[R_A] &= [R_A][R_{L2}][R_B][R_B]^{-1}[R_{L1}]^{-1}[R_A]^{-1}[R_A] \\ &= [R_A][R_{L2}][R_{L1}]^{-1} \\ &= [R_A][L] \end{aligned} \tag{44}$$

where

$$[L] = [R_{L2}][R_{L1}]^{-1} = \begin{bmatrix} L_2^+ L_1^- & 0 \\ 0 & L_1^+ L_2^- \end{bmatrix} = \begin{bmatrix} L^+ & 0 \\ 0 & L^- \end{bmatrix} \tag{45}$$

Defining

$$[P] = [R_{D2}][R_{D1}]^{-1}$$

equation (44) becomes

$$[P][R_A] = [R_A][L]$$

or

$$\begin{bmatrix} P_{11}R_{A11} + P_{12}R_{A21} & P_{11}R_{A12} + P_{12}R_{A22} \\ P_{21}R_{A11} + P_{22}R_{A21} & P_{21}R_{A12} + P_{22}R_{A22} \end{bmatrix} = \begin{bmatrix} L^+ R_{A11} & L^- R_{A12} \\ L^+ R_{A21} & L^- R_{A22} \end{bmatrix} \tag{46}$$

Solve for the ratios below using (39) and (46):

$$\frac{R_{A11}}{R_{A21}} = \frac{-P_{12}}{P_{11} - L^+} = \frac{L^+ - P_{22}}{P_{21}} = \frac{\Delta}{S_{22A}} = S_{11A} - \frac{S_{12A}S_{21A}}{S_{22A}} \tag{47}$$

$$\frac{R_{A12}}{R_{A22}} = \frac{-P_{12}}{P_{11} - L^-} = \frac{L^- - P_{22}}{P_{21}} = S_{11A} \tag{48}$$

Eliminating $[R_A]$ from (40) and (41) following a procedure similar to that above, one obtains,

$$[R_B][Q] = [L][R_B]$$

where $[Q] = [R_{D1}]^{-1}[R_{D2}]$, and

$$\frac{R_{B11}}{R_{B12}} = \frac{-Q_{21}}{Q_{11} - L^+} = \frac{L^+ - Q_{22}}{Q_{12}} = -\frac{\Delta}{S_{11B}} = -S_{22B} + \frac{S_{12B}S_{21B}}{S_{11B}} \tag{49}$$

$$\frac{R_{B21}}{R_{B22}} = \frac{-Q_{21}}{Q_{11} - L^-} = \frac{L^- - Q_{22}}{Q_{12}} = -S_{22B} \tag{50}$$

From (47)-(50),

$$
\begin{aligned}
(L^+)^2 - (P_{11} + P_{22})L^+ + \Delta P &= 0 & (51)\\
(L^-)^2 - (P_{11} + P_{22})L^- + \Delta P &= 0 & (52)\\
(L^+)^2 - (Q_{11} + Q_{22})L^+ + \Delta Q &= 0 & (53)\\
(L^-)^2 - (Q_{11} + Q_{22})L^- + \Delta Q &= 0 & (54)
\end{aligned}
$$

where $\Delta P = P_{11}P_{22} - P_{12}P_{21}$ and $\Delta Q = Q_{11}Q_{22} - Q_{12}Q_{21}$. Note that

$$\Delta P = [R_{D2}][R_{D1}]^{-1} = [R_{D1}]^{-1}[R_{D2}] = \Delta Q$$

Subtracting (53) from (51) or (54) from (52) implies,

$$P_{11} + P_{22} = Q_{11} + Q_{22}$$

Therefore, L^+ and L^- are the two roots of the quadratic equation

$$(L^{\pm})^2 - TrPL^{\pm} + \Delta P = 0 \qquad (55)$$

In the idealized case where there are no losses, L^+ and L^- form a conjugate pair of roots.

Equation (55) can be solved and the ratios (47)-(50) evaluated if the elements of $[P]$ and $[Q]$ are known. These are determined from the NWA s-parameter measurements made by successively connecting line 1 and line 2 between error boxes A and B. This procedure is illustrated in Appendix A.

Now insert a short with unknown reflection coefficient, γ, at each reference plane, as shown in Figure 3 and Figure 4.

From the diagram in Figure 3,

$$\begin{bmatrix} \rho_A \\ 1 \end{bmatrix} = \begin{bmatrix} R_{A11} & R_{A12} \\ R_{A21} & R_{A22} \end{bmatrix} \begin{bmatrix} \gamma b_2 \\ b_2 \end{bmatrix}$$

Therefore,

$$\rho_A = (\gamma R_{A11} + R_{A12})b_2 \qquad (56)$$

$$1 = (\gamma R_{A21} + R_{A22})b_2 \qquad (57)$$



Figure 3: Reflect at reference plane of DUT



Figure 4: Reflect at reference plane of DUT

Eliminating b_2 from (56) and (57), one obtains

$$\rho_A(\gamma R_{A21} + R_{A22}) = \gamma R_{A11} + R_{A12} \tag{58}$$

From the diagram in Figure 4,

$$\begin{bmatrix} b_1 \\ \gamma b_1 \end{bmatrix} = \begin{bmatrix} R_{B11} & R_{B12} \\ R_{B21} & R_{B22} \end{bmatrix} \begin{bmatrix} 1 \\ \rho_B \end{bmatrix}$$

Therefore,

$$b_1 = R_{B11} + \rho_B R_{B22} \tag{59}$$

$$\gamma b_1 = R_{B21} + \rho_B R_{B22} \tag{60}$$

Eliminating b_1 from (59) and (60), one obtains

$$\gamma(R_{B11} + \rho_B R_{B22}) = R_{B21} + \rho_B R_{B22} \tag{61}$$

Eliminating γ from (58) and (61), one obtains

$$\alpha \frac{R_{A22}}{R_{A21}} = \beta \frac{R_{B22}}{R_{B12}} \tag{62}$$

where

$$\alpha = \frac{\frac{R_{A12}}{R_{A22}} - \rho_A}{\rho_A - \frac{R_{A11}}{R_{A21}}} \tag{63}$$

and

$$\beta = \frac{\frac{R_{B21}}{R_{B22}} + \rho_B}{\rho_B + \frac{R_{B11}}{R_{B12}}} \tag{64}$$

Note that ρ_A and ρ_B are known from the NWA measurements made by successively inserting short 1 and short 2 at each reference plane.

Consider the NWA measurement with line 1 inserted. The reflection coefficient for this measurement is:

$$\begin{aligned} S_{11D1} &\doteq \frac{R_{D1}^{12}}{R_{D1}^{22}} = \frac{R_{A11}R_{B12}L_1^+ + R_{A12}R_{B22}L_1^-}{R_{A21}R_{B12}L_1^+ + R_{A22}R_{B22}L_1^-} \\ &= \frac{\frac{R_{A11}}{R_{A21}} + \frac{R_{A12}}{R_{A21}}\frac{R_{B22}}{R_{B12}}\frac{L_1^-}{L_1^+}}{1 + \frac{R_{A22}}{R_{A21}}\frac{R_{B22}}{R_{B12}}\frac{L_1^-}{L_1^+}} \end{aligned} \tag{65}$$

From (62),

$$\frac{R_{B22}}{R_{B12}} = \frac{\alpha}{\beta}\frac{R_{A22}}{R_{A21}}$$

Substituting in (65), one obtains

$$\begin{aligned} S_{11D1} &= \frac{\frac{R_{A11}}{R_{A21}} + \left(\frac{\alpha}{\beta}\right)\left(\frac{L_1^-}{L_1^+}\right)\frac{R_{A12}}{R_{A21}}\frac{R_{A22}}{R_{A21}}}{1 + \left(\frac{\alpha}{\beta}\right)\left(\frac{L_1^-}{L_1^+}\right)\frac{R_{A22}}{R_{A21}}\frac{R_{A22}}{R_{A21}}} \\ &= \frac{\frac{R_{A11}}{R_{A21}} + \left(\frac{\alpha}{\beta}\right)\left(\frac{L_1^-}{L_1^+}\right)\frac{R_{A12}}{R_{A22}}\left(\frac{R_{A11}}{R_{A21}}\right)^2\left(\frac{R_{A22}}{R_{A11}}\right)^2}{1 + \left(\frac{\alpha}{\beta}\right)\left(\frac{L_1^-}{L_1^+}\right)\left(\frac{R_{A11}}{R_{A21}}\right)^2\left(\frac{R_{A22}}{R_{A11}}\right)^2} \end{aligned} \tag{66}$$

From (66) the ratio $\frac{R_{A22}}{R_{A11}}$ can be determined

$$\frac{R_{A22}}{R_{A11}} = \pm\sqrt{\frac{\frac{1}{S_{11D1}}\frac{R_{A11}}{R_{21}} - 1}{(1 - \frac{1}{S_{11D1}}\frac{R_{A12}}{R_{A22}})(\frac{\alpha}{\beta})(\frac{L_1^-}{L_1^+})(\frac{R_{A11}}{R_{A21}})^2}} \tag{67}$$

Using (62) the ratio $\frac{R_{B22}}{R_{B11}}$ can be determined

$$\frac{R_{B22}}{R_{B11}} = \frac{\frac{R_{B22}}{R_{B12}}}{\frac{R_{B11}}{R_{B12}}} = \frac{\frac{\alpha}{\beta}\frac{R_{A22}}{R_{A21}}}{\frac{R_{B11}}{R_{B12}}} = \frac{\alpha}{\beta}\frac{\frac{R_{A11}}{R_{A21}}}{\frac{R_{B11}}{R_{B12}}}\frac{R_{A22}}{R_{A11}} \tag{68}$$

In order to evaluate (67), (68) and the error terms in (34)-(37), the values of L_1^+ and L_1^- must be calculated. Defining

$$\xi \doteq \frac{L_2}{L_1} \quad where \quad L_2 > L_1$$

as the ratio of the two lengths of transmission line, it is shown in Appendix B that

$$L_1^- = \left(L^-\right)^{\frac{1}{(\xi-1)}}$$

and

$$L_1^+ = \left(L^+\right)^{\frac{1}{(\xi-1)}}$$

where L^+ and L^- are the two roots of (55).

It is also necessary to select the proper root when evaluating (67) and (68). This is accomplished by estimating the phase of the reflect, as shown in Appendix C. Physically, the selection corresponds to distinguishing between an open or a short at the reference plane of the DUT.

Using (39),

$$S_{11A} = S_{21A} R_{A12} = \frac{R_{A12}}{R_{A22}} \tag{69}$$

$$S_{22A} = -S_{21A} R_{A21} = -\frac{R_{A21}}{R_{A22}} = -\frac{R_{A21}}{R_{A11}} \frac{R_{A11}}{R_{A22}} \tag{70}$$

and

$$S_{12A} S_{21A} = S_{21A} R_{A11} + S_{11A} S_{22A} = S_{11A} S_{22A} + \frac{R_{A11}}{R_{A22}} \tag{71}$$

From (69), (70) and (71), one obtains

$$\begin{aligned} S_{12A} S_{21A} &= \frac{R_{A12}}{R_{A22}} \left(-\frac{R_{A21}}{R_{A11}} \frac{R_{A11}}{R_{A22}} \right) + \frac{R_{A11}}{R_{A22}} \\ &= \frac{R_{A11}}{R_{A22}} - \frac{R_{A11}}{R_{A22}} \frac{R_{A12}}{R_{A22}} \frac{R_{A21}}{R_{A11}} \end{aligned} \tag{72}$$

Similarly,

$$S_{11B} = \frac{R_{B12}}{R_{B11}} \frac{R_{B11}}{R_{B22}} \tag{73}$$

$$S_{22B} = -\frac{R_{B21}}{R_{B22}} \tag{74}$$

and

$$S_{21B} S_{12B} = \frac{R_{B11}}{R_{B22}} - \frac{R_{B11}}{R_{B22}} \frac{R_{B21}}{R_{B22}} \frac{R_{B12}}{R_{B11}} \tag{75}$$

The transmission coefficient for the NWA measurement with line 1 inserted is given by

$$\begin{aligned} S_{21D1} \doteq \frac{1}{R_{D1}^{22}} &= \frac{1}{R_{A21} R_{B12} L_1^+ + R_{A22} R_{B22} L_1^-} \\ &= \frac{1}{R_{A21} R_{B12} L_1^+ \left[1 + \frac{R_{A22}}{R_{A21}} \frac{R_{B22}}{R_{B12}} \left(\frac{L_1^-}{L_1^+} \right) \right]} \end{aligned} \tag{76}$$

From (39) and (62),

$$
\begin{aligned}
S_{21D1} &= \frac{1}{-\frac{S_{22A}}{S_{21A}}\frac{S_{11B}}{S_{21B}}L_1^+\left[1+\left(\frac{\alpha}{\beta}\right)\left(\frac{L_1^-}{L_1^+}\right)\left(\frac{R_{A22}}{R_{A21}}\right)^2\right]} \\
&= \frac{S_{21A}S_{21B}}{-S_{22A}S_{11B}L_1^+}\frac{1}{\left[1+\left(\frac{\alpha}{\beta}\right)\left(\frac{L_1^-}{L_1^+}\right)\left(\frac{R_{A11}}{R_{A21}}\right)^2\left(\frac{R_{A22}}{R_{A11}}\right)^2\right]}
\end{aligned}
\tag{77}
$$

From(77),

$$
S_{21A}S_{21B} = -S_{21D1}S_{22A}S_{11B}L_1^+\left[1+\left(\frac{\alpha}{\beta}\right)\left(\frac{L_1^-}{L_1^+}\right)\left(\frac{R_{A11}}{R_{A21}}\right)^2\left(\frac{R_{A22}}{R_{A11}}\right)^2\right] \tag{78}
$$

From (40),

$$
[R_{D1}] = [R_A]\,[R_{L1}]\,[R_B]
$$

Therefore,

$$
|R_{D1}| = |R_A||R_{L1}||R_B| = \frac{S_{12A}}{S_{21A}}\frac{S_{12B}}{S_{21B}}L_1^+L_1^- \tag{79}
$$

By definition,

$$
L_1^+L_1^- = L_1^+\frac{1}{L_1^+} = 1 \tag{80}
$$

From (79) and (80),

$$
S_{12A}S_{12B} = |R_{D1}|S_{21A}S_{21B} = \frac{S_{12D1}}{S_{21D1}}S_{21A}S_{21B} \tag{81}
$$

From (78) and (81),

$$
S_{12A}S_{12B} = -S_{12D1}S_{22A}S_{11B}L_1^+\left[1+\left(\frac{\alpha}{\beta}\right)\left(\frac{L_1^-}{L_1^+}\right)\left(\frac{R_{A11}}{R_{A21}}\right)^2\left(\frac{R_{A22}}{R_{A11}}\right)^2\right] \tag{82}
$$

From (69)-(75), (70) and (82) the relevant s-parameters of error networks A and B are expressed in terms of known ratios, and therefore the error terms in equations (34)-(37) can be evaluated.

5 De-Embedding the S-Parameters of the Test Device

The purpose of this section is to summarize the results obtained thus far, and give the expressions which show how to de-embed the s-parameters of the DUT from the NWA measurements.

Equations (34)-(37) are expressed in the form

$$A_1 = A_2 S_{11} + A_3 S_{12} \tag{83}$$

$$B_1 = B_2 S_{21} + B_3 S_{22} \tag{84}$$

$$C_1 = C_2 S_{22} + C_3 S_{21} \tag{85}$$

$$D_1 = D_2 S_{12} + D_3 S_{11} \tag{86}$$

where

$$\begin{aligned}
A_1 &= (S_{11m} - E_{df})E_{tf} \\
A_2 &= (E_{rf} + E_{sf}S_{11m} - E_{df}E_{sf})E_{tf} \\
A_3 &= S_{21m}E_{rf}E_{lf}
\end{aligned}$$

$$\begin{aligned}
B_1 &= S_{21m}E_{rf} \\
B_2 &= (E_{rf} + E_{sf}S_{11m} - E_{df}E_{sf})E_{tf} \\
B_3 &= S_{21m}E_{rf}E_{lf}
\end{aligned}$$

$$\begin{aligned}
C_1 &= (S_{22m} - E_{dr})E_{tr} \\
C_2 &= (E_{rr} + E_{sr}S_{22m} - E_{dr}E_{sr})E_{tr} \\
C_3 &= S_{12m}E_{rr}E_{lr}
\end{aligned}$$

$$\begin{aligned}
D_1 &= S_{12m}E_{rr} \\
D_2 &= (E_{rr} + E_{sr}S_{22m} - E_{dr}E_{sr})E_{tr} \\
D_3 &= S_{12m}E_{rr}E_{lr}
\end{aligned}$$

The s-parameters of the test device are obtained by solving the linear, simultaneous equations (83)-(86) to yield,

$$S_{11} = \frac{A_1 D_2 - D_1 A_3}{A_2 D_2 - D_3 A_3}$$

$$S_{12} = \frac{A_2 D_1 - D_3 A_1}{A_2 D_2 - D_3 A_3}$$
$$S_{21} = \frac{B_1 C_2 - C_1 B_3}{B_2 C_2 - C_3 B_3}$$
$$S_{22} = \frac{B_2 C_1 - C_3 B_1}{B_2 C_2 - C_3 B_3}$$

A listing of the FORTRAN code to implement the generalized TRL algorithm is given in Appendix D.

6 Calculation of Impedance from De-Embedded S-Parameters

It is worthwhile to note the relationship between the s-parameters just found, and the concept of shunt impedance which is usually applied to beamline components.

The relation between the scattering matrix, [S], and the impedance matrix, [Z], for the DUT is given by [5],

$$[S] = ([Z] + [I])^{-1}([Z] - [I]) \tag{87}$$

From (87),

$$([Z] + [I])[S] = [Z] - [I]$$
$$[Z] - [Z][S] = [S] + [I]$$
$$[Z]([I] - [S]) = [S] + [I]$$

and therefore

$$[Z] = ([S] + [I])([I] - [S])^{-1} \tag{88}$$

Consider the equivalent networks



and



Assuming shunt losses are not negligible, interpret the longitudinal impedance as

$$\frac{Z}{Z_0} = \frac{1}{Y_{12}}$$

By definition, the admittance matrix, [Y], is given by

$$[Y] = [Z]^{-1} \tag{89}$$

From (88) and (89),

$$\begin{aligned} [Y] &= ([I] - [S])([S] + [I])^{-1} \\ &= \frac{1}{1 + S_{11} + S_{22} + \Delta} \begin{bmatrix} 1 - S_{11} & -S_{12} \\ -S_{21} & 1 - S_{22} \end{bmatrix} \cdot \\ &\quad \begin{bmatrix} S_{22} + 1 & -S_{12} \\ -S_{21} & S_{11} + 1 \end{bmatrix} \end{aligned} \tag{90}$$

where $\Delta = S_{11}S_{22} - S_{12}S_{21}$.
From (90),

$$Y_{12} = \frac{2S_{12}}{1 + S_{11} + S_{22} + \Delta} = \frac{2S_{21}}{1 + S_{11} + S_{22} + \Delta} = Y_{21}$$

Therefore, the impedance, Z, is given by

$$Z = \frac{Z_0(1 + S_{11} + S_{22} + \Delta)}{2S_{21}} \tag{91}$$

A listing of the FORTRAN code to implement (91) is given in Appendix E.

7 Conclusions

An algorithm has been derived for de-embedding the impedance parameters of a general 2-port network from a realistic set of s-parameter measurements including the effects of external impedance transformations. The method requires the separate measurement of inserted delays of two different lengths (optimally different by $\lambda/2$), and the measurement of identical, but possibly nonideal, reflects. Moreover, the algorithm has been implemented in the form of a FORTRAN computer code, which can be used with standard NWA output data to provide comparatively accurate values for the de-embedded impedance of a given device over as much as an octave in frequency. This method has the advantage of properly taking into account the often-experienced nonideal transmission line standards encountered in these measurements. The details of the comparison of this algorithm with synthesized data, as well as with an actual device whose impedance is known theoretically, are covered in a separate document [6].

References

[1] G.F. Engen and C.A. Hoer, "Thru-Reflect-Line: An Improved Technique for Calibrating the Dual Six-Port Automatic Network Analyzer", *IEEE Transactions on Microwave Theory and Techniques*, Vol. MTT-27, No. 12, pp. 987-993, Dec. 1979.

[2] R.R Pantoja, M.J. Howes, J.R. Richardson and R.D. Pollard, "Improved Calibration and Measurement of the Scattering Parameters of Microwave Integrated Circuits", *IEEE Transactions on Microwave Theory and Techniques*, Vol. MTT-37, No. 11, pp. 1675-1680, Nov. 1989.

[3] J. Staudinger and W. Seely, "MMIC Tests Improved with Standards on Chip", *Microwaves and RF*, pp. 107-114, Feb. 1987.

[4] D. McGinnis, *Thru-Short-Delay De-Embedding*, internal communication, Fermi National Accelerator Laboratory, April 1991.

[5] R.E. Collin, *Field Theory of Guided Waves*, 2nd Edition, IEEE Press, pp. 191-192, 1991.

[6] E. Barsotti, Jr., P. Colestock, and M. Foley, "A Comparison of Thru-Short-Delay and Thru-Reflect-Line De-Embedding", *FNAL Technical Memo*, TM-1782, April 1993.



Figure 5: Calibration Networks D1 and D2

Appendix

A Evaluation of P and Q Matrices

Consider the networks D1 and D2 which are formed by successively connecting transmission line 1 and transmission line 2 between error networks A and B, as shown in Figure 5.

NWA measurements on D1 and D2 yield the s-parameter matrices $[S_{D1}]$ and

$[S_{D2}]$ respectively. Using (39), the corresponding wave cascade matrices are generated,

$$[R_{D1}] = \frac{1}{S_{21D1}} \begin{bmatrix} -\Delta_{D1} & S_{11D1} \\ -S_{22D1} & 1 \end{bmatrix}$$

$$[R_{D2}] = \frac{1}{S_{21D2}} \begin{bmatrix} -\Delta_{D2} & S_{11D2} \\ -S_{22D2} & 1 \end{bmatrix}$$

where $\Delta_{D1} = S_{11D1}S_{22D1} - S_{12D1}S_{21D1}$ and $\Delta_{D2} = S_{11D2}S_{22D2} - S_{12D2}S_{21D2}$.

$$[R_{D1}]^{-1} = \frac{1}{S_{12D1}} \begin{bmatrix} 1 & -S_{11D1} \\ S_{22D1} & -\Delta_{D1} \end{bmatrix}$$

The elements of [P] are obtained from

$$[P] = [R_{D2}]\,[R_{D1}]^{-1} = \frac{1}{S_{21D2}} \begin{bmatrix} -\Delta_{D2} & S_{11D2} \\ -S_{22D2} & 1 \end{bmatrix} \frac{1}{S_{12D1}} \begin{bmatrix} 1 & -S_{11D1} \\ S_{22D1} & -\Delta_{D1} \end{bmatrix}$$

$$= \frac{1}{S_{12D1}S_{21D2}} \begin{bmatrix} (S_{11D2}S_{22D1} - \Delta_{D2}) & (S_{11D1}\Delta_{D2} - S_{11D2}\Delta_{D1}) \\ (S_{22D1} - S_{22D2}) & (S_{11D1}S_{22D2} - \Delta_{D1}) \end{bmatrix}$$

Similarly,

$$[Q] = [R_{D1}]^{-1}\,[R_{D2}] = \frac{1}{S_{12D1}} \begin{bmatrix} 1 & -S_{11D1} \\ S_{22D1} & -\Delta_{D1} \end{bmatrix} \frac{1}{S_{12D2}} \begin{bmatrix} -\Delta_{D2} & S_{11D2} \\ -S_{22D2} & 1 \end{bmatrix}$$

$$= \frac{1}{S_{12D1}S_{21D2}} \begin{bmatrix} (S_{11D1}S_{22D2} - \Delta_{D2}) & (S_{11D2} - S_{11D1}) \\ (S_{22D2}\Delta_{D1} - S_{22D1}\Delta_{D2}) & (S_{11D2}S_{22D1} - \Delta_{D1}) \end{bmatrix}$$

B Calculation of L_1^+ and L_1^-

Define ξ as the ratio of the lengths of the two transmission line calibration standards.

$$\xi \doteq \frac{L_2}{L_1} \quad where \quad L_2 > L_1$$

By definition $L^+ = L_2^+ L_1^-$ and $L^- = L_1^+ L_2^-$. Therefore,

$$\begin{aligned} L^+ &= e^{-a(L_2-L_1)} e^{-jk(L_2-L_1)} \\ &= e^{-(a+jk)(L_2-L_1)} \\ &= e^{-\sigma(L_2-L_1)} \end{aligned} \tag{92}$$

and

$$L^- = \frac{1}{L^+} = e^{\sigma(L_2-L_1)} \tag{93}$$

where $\sigma = (a + jk), a > 0$ is the complex propagation constant for the two transmission line standards. From (92) and (93),

$$\begin{aligned} L_1^+ &= e^{-\sigma L_1} \\ &= e^{-\sigma(L_2-L_1)/\frac{(L_2-L_1)}{L_1}} \\ &= \left(L^+\right)^{\frac{1}{(\xi-1)}} \end{aligned}$$

Similarly,

$$\begin{aligned} L_1^- &= e^{\sigma L_1} \\ &= e^{\sigma(L_2-L_1)/\frac{(L_2-L_1)}{L_1}} \\ &= \left(L^-\right)^{\frac{1}{(\xi-1)}} \end{aligned}$$

Therefore,

$$L_1^+ = \left(L^+\right)^{\frac{1}{(\xi-1)}}$$

$$L_1^- = \left(L^-\right)^{\frac{1}{(\xi-1)}}$$

C Proper Root Choice

The proper choice of root in evaluating $\frac{R_{A22}}{R_{A11}}$ is accomplished by estimating the phase of the reflect, as shown below.

From (58),

$$\rho_A = \frac{\gamma R_{A11} + R_{A12}}{\gamma R_{A21} + R_{A22}} = \frac{\gamma \frac{R_{A11}}{R_{A22}} + \frac{R_{A12}}{R_{A22}}}{\gamma \frac{R_{A21}}{R_{A22}} + 1} \tag{94}$$

From (94),

$$\gamma \frac{R_{A11}}{R_{A22}} + \frac{R_{A12}}{R_{A22}} = \gamma \rho_A \frac{R_{A21}}{R_{A22}} + \rho_A$$

and

$$\gamma \left(\frac{R_{A11}}{R_{A22}} - \rho_A \frac{R_{A21}}{R_{A22}} \right) = \rho_A - \frac{R_{A12}}{R_{A22}}$$

Therefore, the unknown reflection coefficient, γ, is given by

$$\gamma = \frac{\rho_A - \frac{R_{A12}}{R_{A22}}}{\frac{R_{A11}}{R_{A22}} \left[1 - \rho_a \left(\frac{\frac{R_{A21}}{R_{A22}}}{\frac{R_{A11}}{R_{A22}}} \right) \right]} = \frac{\frac{R_{A22}}{R_{A11}} \left[\rho_A - \frac{R_{A12}}{R_{A22}} \right]}{\left[1 - \rho_A \left(1 / \frac{R_{A11}}{R_{A21}} \right) \right]} \tag{95}$$

Using (47), (48) and the value of ρ_A from NWA measurements, γ can be evaluated from (95) for each choice of root in (67). Since γ represents the reflection coefficient for a short, the proper root choice is that value of $\frac{R_{A22}}{R_{A11}}$ for which the corresponding value of γ lies in the shaded region of the complex plane.

γ - Plane



D TRL Algorithm FORTRAN Code

```
*************************************************************************
*                                                                       *
*                             F E R M I                                 *
*                                                                       *
*       N A T I O N A L   A C C E L E R A T O R   L A B O R A T O R Y   *
*                                                                       *
*       The purpose of D_EMBED is to extract the s-parameters of a      *
*       test device from NWA measurements. D_EMBED implements the       *
*       generalized TRL algorithm. See Fermi National Accelerator       *
*       Laboratory Technical Memo No. 1781  for the theoretical         *
*       development.                                                    *
*                                                                       *
*       Author: Michael Foley                       708/840-2505        *
*               AD/Mechanical Engineering Support                       *
*                                                                       *
*************************************************************************

        program d_embed

        common/spard/sl1(2,2),sl2(2,2),ss1(2,2),ss2(2,2),sm(2,2)
        common/sparc/s(2,2)
        complex*16 sl1,sl2,ss1,ss2,sm,s
        real*8 freq,ratio
        integer npoints,data_format

c       Open the input data files and an output file

        call file_open(ratio,npoints,data_format)

c       De-embed the s-parameters of the test device

        do 11 i=1,npoints
            call file_read(freq,data_format)
            call s_parameter(freq,ratio)
            write(96,100) freq,s(1,1),s(2,1),s(1,2),s(2,2)
11      continue

        close(unit=91)
        close(unit=92)
        close(unit=93)
        close(unit=94)
        close(unit=95)
        close(unit=96)

100     format(1p9e12.4)

        stop
        end

        subroutine file_open(ratio,npoints,data_format)
        character*20 file_line1,file_line2,file_short1,
     x                  file_short2,file_data,file_output
        real*8 ratio
        integer data_format
        character*100 title(22)

        type 11
11      format(' ','Enter the name of the Line 1 calibration file (use single
     x quotes to enclose the file name):',$)
        accept*,file_line1
```

```
      type 12
12    format(' ','Enter the name of the Line 2 calibration file (use single
     x quotes to enclose the file name):',$)
      accept*,file_line2

      type 13
13    format(' ','Enter the name of the Short 1 calibration file (use singl
     xe quotes to enclose    the file name):',$)
      accept*,file_short1

      type 14
14    format(' ','Enter the name of the Short 2 calibration file (use singl
     xe quotes to enclose    the file name):',$)
      accept*,file_short2

      type 15
15    format(' ','Enter the name of the file containing the measured s-para
     xmeters for the test    device (use single quotes):',$)
      accept*,file_data

      type 16
16    format(' ','Enter the name of the output file for the computed s-para
     xmeters for the test    device (use single quotes):',$)
      accept*,file_output

      type 17
17    format(' ','Enter the number of data points in the files:',$)
      accept*,npoints

      type 18
18    format(' ','Enter the ratio of the lengths of the two transmission li
     xnes (L2/L1) - If Line1 is a direct connection enter 0:',$)
      accept*,ratio

      type 19
19    format(' ','Is calibration file data in mag,arg format(1) or re,im
     xformat(2):',$)
      accept*,data_format

      open(unit=91,file=file_line1,status='old')
      open(unit=92,file=file_line2,status='old')
      open(unit=93,file=file_short1,status='old')
      open(unit=94,file=file_short2,status='old')
      open(unit=95,file=file_data,status='old')
      open(unit=96,file=file_output,status='new')

c     Read data file headers

      do 20 i=1,22
          read(91,100) title(i)
          read(92,100) title(i)
          read(93,100) title(i)
          read(94,100) title(i)
          read(95,100) title(i)
20    continue

100   format(a100)

      return
      end
```

```
      subroutine file_read(freq,data_format)
      common/spard/s11(2,2),s12(2,2),ss1(2,2),ss2(2,2),sm(2,2)
      complex*16 s11,s12,ss1,ss2,sm
      real*8 freq
      real*8 s11m,s11a,s21m,s21a,s12m,s12a,s22m,s22a
      integer data_format

      if (data_format.eq.1) then

c     Data in magnitude,argument format

      read(91,*) freq,s11m,s11a,s21m,s21a,s12m,s12a,s22m,s22a
      s11(1,1)=dcmplx(s11m*dcosd(s11a),s11m*dsind(s11a))
      s11(2,1)=dcmplx(s21m*dcosd(s21a),s21m*dsind(s21a))
      s11(1,2)=dcmplx(s12m*dcosd(s12a),s12m*dsind(s12a))
      s11(2,2)=dcmplx(s22m*dcosd(s22a),s22m*dsind(s22a))

      read(92,*) freq,s11m,s11a,s21m,s21a,s12m,s12a,s22m,s22a
      s12(1,1)=dcmplx(s11m*dcosd(s11a),s11m*dsind(s11a))
      s12(2,1)=dcmplx(s21m*dcosd(s21a),s21m*dsind(s21a))
      s12(1,2)=dcmplx(s12m*dcosd(s12a),s12m*dsind(s12a))
      s12(2,2)=dcmplx(s22m*dcosd(s22a),s22m*dsind(s22a))

      read(93,*) freq,s11m,s11a,s21m,s21a,s12m,s12a,s22m,s22a
      ss1(1,1)=dcmplx(s11m*dcosd(s11a),s11m*dsind(s11a))
      ss1(2,1)=dcmplx(s21m*dcosd(s21a),s21m*dsind(s21a))
      ss1(1,2)=dcmplx(s12m*dcosd(s12a),s12m*dsind(s12a))
      ss1(2,2)=dcmplx(s22m*dcosd(s22a),s22m*dsind(s22a))

      read(94,*) freq,s11m,s11a,s21m,s21a,s12m,s12a,s22m,s22a
      ss2(1,1)=dcmplx(s11m*dcosd(s11a),s11m*dsind(s11a))
      ss2(2,1)=dcmplx(s21m*dcosd(s21a),s21m*dsind(s21a))
      ss2(1,2)=dcmplx(s12m*dcosd(s12a),s12m*dsind(s12a))
      ss2(2,2)=dcmplx(s22m*dcosd(s22a),s22m*dsind(s22a))

      read(95,*) freq,s11m,s11a,s21m,s21a,s12m,s12a,s22m,s22a
      sm(1,1)=dcmplx(s11m*dcosd(s11a),s11m*dsind(s11a))
      sm(2,1)=dcmplx(s21m*dcosd(s21a),s21m*dsind(s21a))
      sm(1,2)=dcmplx(s12m*dcosd(s12a),s12m*dsind(s12a))
      sm(2,2)=dcmplx(s22m*dcosd(s22a),s22m*dsind(s22a))

      else if (data_format.eq.2) then

c     Data in real,imaginary format

      read(91,99) freq,s11(1,1),s11(2,1),s11(1,2),s11(2,2)
      read(92,99) freq,s12(1,1),s12(2,1),s12(1,2),s12(2,2)
      read(93,99) freq,ss1(1,1),ss1(2,1),ss1(1,2),ss1(2,2)
      read(94,99) freq,ss2(1,1),ss2(2,1),ss2(1,2),ss2(2,2)
      read(95,99) freq,sm(1,1),sm(2,1),sm(1,2),sm(2,2)

      endif

99    format(9e12.4)

      return
      end

      subroutine s_parameter(freq,ratio)
      implicit complex*16 (a-h,o-z)
```

```
        common/spard/sl1(2,2),sl2(2,2),ss1(2,2),ss2(2,2),sm(2,2)
        common/sparc/s(2,2)
        dimension p(2,2),q(2,2)
        complex*16 sl1,sl2,ss1,ss2,sm
        complex*16 p,q,s,lplus,lminus,gamma,z
        real*8 freq,ratio,power,garg
        real*8 lpmag,lparg

c       Evaluate P and Q matrices

        deltasl1=sl1(1,1)*sl1(2,2)-sl1(2,1)*sl1(1,2)
        deltasl2=sl2(1,1)*sl2(2,2)-sl2(2,1)*sl2(1,2)

        p(1,1)=sl2(1,1)*sl1(2,2)-deltasl2
        p(2,1)=sl1(2,2)-sl2(2,2)
        p(1,2)=sl1(1,1)*deltasl2-sl2(1,1)*deltasl1
        p(2,2)=sl1(1,1)*sl2(2,2)-deltasl1

        q(1,1)=sl1(1,1)*sl2(2,2)-deltasl2
        q(2,1)=sl2(2,2)*deltasl1-sl1(2,2)*deltasl2
        q(1,2)=sl2(1,1)-sl1(1,1)
        q(2,2)=sl2(1,1)*sl1(2,2)-deltasl1

        do 11 i=1,2
            do 11 j=1,2
                p(i,j)=(1./(sl1(1,2)*sl2(2,1)))*p(i,j)
                q(i,j)=(1./(sl1(1,2)*sl2(2,1)))*q(i,j)
11      continue

c
c       Calculate the pair of roots (lplus and lminus) of
c       the quadratic equation
c
c               L**2 - TrP*L + DELTAP = 0
c
        deltap=p(1,1)*p(2,2)-p(2,1)*p(1,2)
        deltaq=q(1,1)*q(2,2)-q(2,1)*q(1,2)
        lplus=0.5*((p(1,1)+p(2,2))+sqrt((p(1,1)+p(2,2))**2-4.*deltap))
        lminus=0.5*((p(1,1)+p(2,2))-sqrt((p(1,1)+p(2,2))**2-4.*deltap))

        lpmag=abs(lplus)
        lparg=57.2958*atan2(dimag(lplus),dreal(lplus))

c       Assign roots of quadratic equation to proper location in
c       [L] matrix

        if(lparg.gt.0.) then
            z=lplus
            lplus=lminus
            lminus=z
        endif

************************************************************************
*                                                                      *
*       Note that the code is valid only for  K(L2-L1) < PI. If the    *
*       frequency range is such that  K(L2-L1)  exceeds PI, then the   *
*       code must be modified by creating the appropriate selection    *
*       structure.                                                     *
*                                                                      *
************************************************************************
```

```fortran
c       Calculate wave cascade matrix element ratios

        ra11ra21=(lplus-p(2,2))/p(2,1)
        ra12ra22=(lminus-p(2,2))/p(2,1)
        rb11rb12=(lplus-q(2,2))/q(1,2)
        rb21rb22=(lminus-q(2,2))/q(1,2)
c
c       Evaluate reflections from Short 1 and Short 2 respectively
c
        rho_a=ss1(1,1)
        rho_b=ss2(2,2)

        alpha=(ra12ra22-rho_a)/(rho_a-ra11ra21)
        beta=(rb21rb22+rho_b)/(rho_b+rb11rb12)

c       Calculate L1plus and L1minus

        if(ratio.gt.1) then
                power=1./(ratio-1.)
                lplus=lplus**power
                lminus=lminus**power
        else
                lplus=1.0
                lminus=1.0
        endif

        ra22ra11=-sqrt((ra11ra21*(1./s11(1,1))-1.)/
     1          ((1.-ra12ra22*(1./s11(1,1)))*ra11ra21**2*
     2             (alpha/beta)*(lminus/lplus)))

c       Select proper root by checking phase of reflect

        gamma=ra22ra11*(rho_a-ra12ra22)/(1.-rho_a*(1./ra11ra21))
        garg=57.2958*atan2(dimag(gamma),dreal(gamma))

        if(abs(garg).lt.90.)then
                ra22ra11=-ra22ra11
        endif

        rb22rb11=(alpha/beta)*ra11ra21*ra22ra11/rb11rb12

c       Evaluate appropriate s-parameters of A and B networks

        sa11=ra12ra22
        sa22=-1./(ra11ra21*ra22ra11)
        sa12_sa21=(1./ra22ra11)-ra12ra22/(ra22ra11*ra11ra21)

        sb11=1./(rb11rb12*rb22rb11)
        sb22=-rb21rb22
        sb21_sb12=(1./rb22rb11)-rb21rb22/(rb22rb11*rb11rb12)

        sa21_sb21=-s11(2,1)*sa22*sb11*lplus*
     x         (1.+(alpha/beta)*(lminus/lplus)*(ra11ra21*ra22ra11)**2)
        sa12_sb12=sa21_sb21*(s11(1,2)/s11(2,1))

c       Evaluate error terms for standard NWA error model

        edf=sa11
        esf=sa22*lplus
        erf=sa12_sa21*lplus
```

```
      elf=sb11*lplus
      etf=sa21_sb21*lplus

      edr=sb22
      esr=sb11*lplus
      err=sb21_sb12*lplus
      elr=sa22*lplus
      etr=sa12_sb12*lplus

c     Solve for de-embedded s-parameters of the test device

      a1=(sm(1,1)-edf)*etf
      a2=(erf+esf*sm(1,1)-edf*esf)*etf
      a3=sm(2,1)*erf*elf

      b1=sm(2,1)*erf
      b2=(erf+esf*sm(1,1)-edf*esf)*etf
      b3=sm(2,1)*elf*erf

      c1=(sm(2,2)-edr)*etr
      c2=(err+esr*sm(2,2)-edr*esr)*etr
      c3=sm(1,2)*err*elr

      d1=sm(1,2)*err
      d2=(err+esr*sm(2,2)-edr*esr)*etr
      d3=sm(1,2)*elr*err

      s(1,1)=(a1*d2-d1*a3)/(a2*d2-d3*a3)
      s(1,2)=(a2*d1-d3*a1)/(a2*d2-d3*a3)
      s(2,1)=(b1*c2-c1*b3)/(b2*c2-c3*b3)
      s(2,2)=(b2*c1-c3*b1)/(b2*c2-c3*b3)

      return
      end
```

E Impedance FORTRAN Code

```
      program z_calc

c     Calculate magnitude and phase of impedance from de-imbedded s-parameters

      complex*16 s11(1000),s21(1000),s12(1000),s22(1000),zz(1000)
      complex*16 deltas
      real*8 freq(1000),zmag(1000),zphase(1000)
      real*8 s11m,s11a,s21m,s21a,s12m,s12a,s22m,s22a
      character*20 data_file

      type 8
8     format(' ','Enter name of s-parameter file (single quotes):'$)
      accept*,data_file

      open(unit=91,file=data_file,status='old')
      open(unit=92,file='zmag.dat',status='new')
      open(unit=93,file='zphase.dat',status='new')

      type 11
11    format(' ','Enter number of data points:',$)
      accept*,npoints

      do 12 i=1,npoints

c     Data in real, imaginary format

        read(91,100) freq(i),s11(i),s21(i),s12(i),s22(i)

c     Data in magnitude, phase(degrees) format

c       read(91,*) freq(i),s11m,s11a,s21m,s21a,s12m,s12a,s22m,s22a
c       s11(i)=dcmplx(s11m*dcosd(s11a),s11m*dsind(s11a))
c       s21(i)=dcmplx(s21m*dcosd(s21a),s21m*dsind(s21a))
c       s12(i)=dcmplx(s12m*dcosd(s12a),s12m*dsind(s12a))
c       s22(i)=dcmplx(s22m*dcosd(s22a),s22m*dsind(s22a))

        deltas=s11(i)*s22(i)-s21(i)*s12(i)
        zz(i)=(1.+s11(i)+s22(i)+deltas)/(2.*s21(i))
        zz(i)=266.*zz(i)
12    continue

      do 20 i=1,npoints
          zmag(i)=abs(zz(i))
          zphase(i)=57.2958*atan2(dimag(zz(i)),dreal(zz(i)))
20    continue

      do 40 i=1,npoints
          write(92,*) freq(i),zmag(i)
          write(93,*) freq(i),zphase(i)
40    continue

100   format(9e12.4)

      stop
      end
```